[bookmark: _GoBack]EKSTRA EXCELFUNKSIES
Die volgende notas en oefeninge oor bykomende Excel-funksies is gebaseer op die Graad 12 RTT-eksamenriglyne vir 2016 wat onlangs deur die DBE uitgereik is. As sulks, verteenwoordig dit ’n uitstekende bykomende hulpmiddel vir voorbereiding vir die finale RTT-eksamens. Werk, behalwe vir jou handboek, ook versigtig en deeglik deur hierdie materiaal en vra hulp by jou onderwyser waar nodig. Sterkte vir die eksamen!

[bookmark: _Toc458453762]DATUM- & TYD-FUNKSIES
[bookmark: _Toc456196329]Die volgende Datum- en Tyd-funksies is in die 2016 Eksamenriglyne vir RTT gelys as ’n verdere verduideliking van wat in Vraestel 1 getoets kan word: 
YEAR 
MONTH 
DAY 
DAYS 
DAYS360 
HOUR 
MINUTE 
TIME 
TODAY 
Maak seker dat jy weet hoe om hierdie funksie te gebruik.
[bookmark: _Toc458453763]DATUM-FUNKSIES
Om datum-funksies ten volle te verstaan, is dit belangrik om daarop te let dat Excel datums as opeenvolgende reeksnommers stoor, beginnende by 1900/01/01 (dag 1). Byvoorbeeld, 2016/07/13 is 42 564 dae na 1900/01/01. Om dit te illustreer, tik ’n nommer en formateer dit as datum en omgekeerd. 
Ons begin deur na sommige datum-funksies afsonderlik te kyk en daana sal ons kyk hoe ons hulle in die praktyk saam kan gebruik. 
Maak die sigblad ExcelExercises oop en werk in die DATE Functions-werkblad.
TODAY
Die TODAY-funksie gee die reeksnommer vir die huidige datum, reeds as datum geformateer.  
	[image: ]
	Let op dat hierdie funksie geen argumente (d.w.s enigiets tussen die hakies) nodig het nie.


AKTIWITEIT 1
Voeg ’n funksie in sel G2 in om die huidige datum te vertoon.
DAY
Die DAY-funksie word gebruik om die dag van die maand, as ’n getal van 1 tot 31, van ’n datum te onttrek. 
	[image: ]
	Let op dat ‘serial_number’ in die tooltip na ’n datum in die Excel date/time-formaat verwys.


AKTIWITEIT 2
Voeg ’n funksie in sel H2 in om die dag van die maand van die datum in sel E2 te onttrek. Kopieer die funksie na al die werknemers. 
MONTH
Die MONTH-funksie word gebruik om die maand van ’n datum, as ’n getal van 1 (Januarie) tot 12 (Desember), te onttrek. Die sintaksis vir die MONTH-funksie is =MONTH(serial_number).    
AKTIWITEIT 3
Voeg ’n funksie in sel I2 in om die nommer van die maand van die datum in sel E2 te onttrek. Kopieer die funksie na al die werknemers.
Year
Die YEAR-funksie word gebruik om die jaar van ’n datum te onttrek, as ’n eenvoudig 4-syfer getal. Die sintaksis vir die YEAR-funksie is =YEAR(serial_number)
AKTIWITEIT 4
Voeg ’n funksie in sel J2 in om die jaar van die datum in sel E2 te onttrek. Kopieer die funksie na al die werknemers.
DAYS
Die DAYS-funksie verskaf die aantal dae tussen twee datums. 
	[image: ]
	Let op na die volgorde van die argumente in die tooltip: die start_date (2e  argument, die ‘kleiner’ datum) word afgetrek van die end_date (1e argument, die ‘groter’ datum). Die funksie in hierdie voorbeeld is dus gelyk aan die formule =E2-C2.


AKTIWITEIT 5
Voeg ’n funksie in sel K2 in om die aantal dae tussen die datums in selle C2 en E2 te bepaal. Kopieer die funksie na die ander werknemers.
DAYS360
Die DAYS360-funksie bereken die aantal dae tussen twee datums, gebaseer op 360 dae per jaar in plaas van die gewone 365.25 dae per jaar. Hierdie funksie word meestal in rekeningkunde gebruik.
	[image: ]
	Die sintaksis vir die DAYS360-funksie is soortgelyk aan dié van die DAYS-funksie, buiten dat daar ’n ekstra opsionele argument is vir metode [method] – gebruik om te spesifiseer of die Amerikaanse of Europese metode in die berekening gebruik moet word. As hierdie argument FALSE is, of weggelaat is, word die Amerikaanse metode gebruik.


AKTIWITEIT 6
Voeg ’n funksie in sel L2 in om die aantal dae tussen die datums in selle C2 en E2 te bepaal, gebaseer op 360 dae jaar. Kopieer die funksie na al die werknemers
DATE [nie in die eksamenriglyne gelys nie, maar ‘n nuttige funksie]
Die DATE-funksie word dikwels saam met die DAY-, MONTH- en YEAR-funksies gebruik. Gebruik dit wanneer jy drie aparte waardes (jaar, maand en dag) moet gebruik om ’n reeksnommer te skep wat Excel as ’n datum kan herken.
	[image: ]
	Jy kan selwaardes vir die jaar, maand en dag (soos in die voorbeeld) gebruik, of jy kan eenvoudig getalle gebruik, soos =DATE(2010,6,30). 
Let op: gebruik altyd 4 syfers vir die jaar.





Hierdie funksie word dikwels gebruik om ’n gegewe aantal jare, maande of dae by ’n bestaande datum te voeg. Veronderstel dat die datum 25 October 2010 is sel E2 gestoor is, en ons moet twee maande en drie dae by hierdie datum voeg. 
	[image: ]
	Dit mag natuurlik lyk om eenvoudig na sel E2 in die DATE-funksie te verwys, maar dit sal ’n foutboodskap lewer. Die probleem is dat die ‘year’- argument ’n viersyfer-getal behoort te wees, die ‘month’-argument ’n getal tussen 1 en 12, en die ‘day’-argument ’n getal tussen 1 en 31, terwyl die waarde wat korrek in sel E2 gestoor is, is 40476 (geformateer as die datum 25 October 2010) – vandaar die fout.

	[image: ]
	Die korrekte manier om hierdie funksie te formuleer is om eers die jaar, daarna die maand en dan die dag uit die datum in sel E2 te onttrek. Die YEAR-funksie onttrek die jaar (2010), die MONTH-funksie die maand (10), en die DAY-funksie die dag (25). Die aantal maande en dae wat hierby gevoeg moet word, word dan eenvoudig net by die ‘month’- en ‘day’-argumente respektiewelik, gevoeg.


AKTIWITEIT 7
Voeg ’n DATE-funksie in sel M2 in om 1 jaar en 1 maand by die datum in sel E2 te voeg. Kopieer die funksie na al die werknemers. 
=DATE(YEAR(E2)+1,MONTH(E2)+1,DAY(E2)) or =DATE(J2+1,I2+1,H2) aangesien die aparte funksies reeds in daardie selle uitgevoer is.
WEEKDAY [nie in die eksamenriglyne gelys nie, maar ‘n nuttige funksie]  
WEEKDAY word gebruik om ’n getal van 1 tot 7 te vertoon om die dag van die week van ’n datum te verskaf, waar 1 ’n Sondag en 7 ’n Saterdag is. 
	[image: ]
	Die sintaksis vir Weekday kan baie eenvoudig wees – die tweede opsionele argument [return_type], hoef nie gebruik te word nie.


AKTIWITEIT 8
Voeg ’n funksie in sel N2 in om te bepaal op watter dag van die week die datum in E2 val. Kopieer die funksie na al die werknemers. 
[bookmark: _Toc458453764]TYD-FUNKSIES
Werk nou in die TIME Functions-werkblad.
Om tyd-funksies ten volle te verstaan, is dit belangrik om te besef dat enige gegewe tyd as ’n (desimale) deel van ’n dag gestoor word. Byvoorbeeld, as 0.25 as tyd geformateer word, word dit vertoon as 06:00 (’n kwart van ’n dag). Wanneer jy die formaat yyyy-mm-dd hh:mm op die getal 42 564.25 toepas, vertoon die getal as 2016-07-13 06:00.
NOW
Die NOW-funksie vertoon die reeks desimale getal vir die huidige datum en tyd, reeds geformateer as ’n datum met ’n tyd.
	[image: ]
	Let op dat hierdie funksie nie enige argumente (d.w.s. enigiets tussen die hakies) vereis nie.


AKTIWITEIT 9
Voeg ’n funksie in sel H1 in om die huidige dag en tyd te vertoon.


SECOND [nie in die eksamenriglyne gelys nie, maar ‘n nuttige funksie]
Die SECOND-funksie word gebruik om die sekondes van ’n tyd, as ’n getal van 1 tot 59, te onttrek.
	[image: ]
	Let daarop dat die ‘serial_number’ in die tooltip na ’n tydstip in die Excel date/time-formaat verwys.


AKTIWITEIT 10
Voeg ’n funksie in sel D2 in om die sekondes van die tyd in sel B2 te onttrek. Kopieer die funksie na al die werknemers. 
MINUTE	
Die MINUTE-funksie word gebruik om die minute van ’n tyd, as ’n getal van 1 tot 59 te onttrek. Die sintaksis vir die MINUTE- funksie is =MINUTE(serial_number).    
AKTIWITEIT 11
Voeg ’n funksie in sel E2 in om die minute van die tyd in sel B2 te onttrek. Kopieer die funksie na al die werknemers.
HOUR
Die HOUR-funksie word gebruik om die ure van ’n tyd, as ’n getal van 1 tot 23 te onttrek. Die sintaksis vir die HOUR-funksie is =HOUR(serial_number)
AKTIWITEIT 12
Voeg ’n funksie in sel F2 in om die ure van die tyd in sel B2 te onttrek. Kopieer die funksie na al die werknemers.
TIME
Die TIME-funksie is soortgelyk aan die DATE-funksie, in die opsig dat dit dikwels gebruik word saam met die HOUR-, MINUTE- en SECOND-funksies, wanneer jy drie aparte waardes wil kombineer (hour, minute en second) om ’n desimale getal te vorm wat Excel as tyd sal herken.
	[image: ]
	Jy kan selverwysings vir die uur-, minute- en sekonde-argumente (soos in die voorbeeld) gebruik, of jy kan eenvoudig net getalle gebruik, soos =TIME(23,55,10). 


Hierdie funksie kan gebruik word om ’n gegewe aantal ure, minute of sekondes by ’n gegewe tyd te voeg.
AKTIWITEIT 13
Voeg ’n TIME-funksie in sel G2 in om 3 ure en 25 minute by die tyd in sel B2 te voeg. Kopieer die funksie na al die werknemers 
=TIME(HOUR(B2)+3,MINUTE(B2)+25,SECOND(B2)) of =TIME(F2+3,E2+25,D2) aangesien die afsonderlike funksies reeds in daardie selle uitgevoer is.


[bookmark: _Toc458453765]OPSOEK-FUNKSIES
Die volgende opsoek- en verwysingsfunksies is in die 2016 Eksamenriglyne vir RTT gelys as ’n verdere verduideliking van wat in Vraestel 1 getoets kan word: 
LOOKUP 
VLOOKUP 
HLOOKUP 
Maak die sigblad ExcelExercises oop en werk in die LOOKUP Functions-werkblad.
[bookmark: _Toc458453766]HLOOKUP-FUNKSIE
Jy behoort alreeds die VLOOKUP (vertical lookup) funksie onder die knie te hê. Hierdie funksie word gebruik om ’n waarde in ’n tabel op te soek en dan ’n waarde vanuit ’n sel in die ooreenstemmende ry te vertoon.
Die HLOOKUP (horizontal lookup)-funksie is soortgelyk aan die VLOOKUP-funksie, buiten dat dit rye in plaas van kolomme gebruik.
	[image: ]
	Ons soek hier die tak waarby Andrew werk, om uit te vind wat sy takkode is.

	E2 (lookup_value) 	is die waarde in die hoof tabel wat ons wil opsoek (sy taknaam).
Q$2:AB$4 (table_array) 	is die opsoektabel (die lys van take met elkeen se takkode).
Let op: Soos met die VLOOKUP-funksie, moet 'n waarde in die HLOOKUP se opsoektabel ooreenstem met jou opsoekwaarde.
2 (row_index_num) 	is die nommer van die ry in die lookup table wat jy wil vertoon (die takkode in die tweede ry)
FALSE ([range_lookup]) 	is ’n opsionele argument wat jou toelaat om te spesifiseer of jy ’n ‘exact match (FALSE)’ of ’n ‘approximate match (TRUE)’ wil hê. 


AKTIWITEIT 14
Voeg ’n funksie in sel G2 in om die bonus vas te stel vir die tak waar Andrew Beach werk. Gebruik die reeks Q2:AB3 as die opsoektabel. Kopieer die funksie na al die werknemers.
=HLOOKUP(E2,Q$2:AB$4,3,FALSE) or =HLOOKUP(E2,$Q$2:$AB$4,3,FALSE)


[bookmark: _Toc458453767]LOOKUP-FUNKSIE
Die LOOKUP-funksie word gebruik wanneer jy in ’n enkele kolom of ry moet soek om ’n waarde te kry van dieselfde posisie in ’n tweede kolom. Dit verskil van die VLOOKUP- en HLOOKUP-funksies deurdat die eerste kolom of ry in die lookup table nie noodwendig die een moet wees wat ’n ‘match’ (ooreenkoms) vir die opsoekwaarde moet bevat nie. Kom ons kyk na ’n voorbeeld.
In hierdie voorbeeld het ons ’n lookup table met inligting oor elke department, watter geskenk hulle sal ontvang en wie verantwoordelik is vir die aankoop van die geskenke. 
	[image: ]
	Let eerstens daarop, dat die LOOKUP-funksie op twee maniere gebruik kan word – die ‘vector form’ (eerste reël van die tooltip) of die ‘array form’ (tweede reël van die tooltip). Die ‘array form’ het uitgedien geraak met die toevoeging van VLOOKUP, daarom gebruik ons slegs die ‘vector form’. In hierdie scenario soek ons Andrew se department op om vas te stel watter persoon verantwoordelik is vir die aankoop van sy geskenk. 

	D2 (lookup_value) 	is die waarde in die hoof tabel wat ons wil vergelyk of opsoek (sy department).
M$20:M$23 (lookup_vector) 	is die lookup column (die kolom wat die departmente bevat).
L$20:L$23 ([result_vector]) 	is die kolom wat die resultaat bevat wat jy soek.
Let op dat die LOOKUP-funksie nie die TRUE/FALSE-opsies vir presiese of geskatte ooreenstemmings bevat nie – alle ooreenkomste is geskat. 


AKTIWITEIT 15
Voeg ’n funksie in sel I2 in om uit te vind watter geskenk Andrew sal ontvang. Gebruik die LOOKUP-tabel. Kopieer die funksie na al die werknemers.
=LOOKUP(D2,M$20:M$23,K$20:K$23) of =LOOKUP(D2,$M$20:$M$23,$K$20:$K$23)


[bookmark: _Toc458453768]COUNTIF(S), SUMIF(S) & AVERAGEIF(S)
Die volgende kom uit die 2016 Eksamenriglyne vir RTT: “Kandidate kan (maar is nie beperk tot nie) die SUMIFS- en COUNTIFS-funksies gebruik om probleme op te los eerder as om bestaande funksies te gebruik om komplekse funksies te skep. Die SUMIFS en COUNTIFS kan gebruik word om vrae te beantwoord.” 
Ons het besluit om die AVERAGEIF(S) funksie hier in te sluit, aangesien dit baie handig is om ’n gemiddelde waarde te bepaal wanneer een of meer kriteria gespesifiseer word. (Let egter daarop, dat hierdie funksies NIE in die eksamenriglyne genoem word nie.)
Maak die sigblad ExcelExercises oop en werk in die IFS Functions-werkblad.
[bookmark: _Toc456196325][bookmark: _Toc458453769]COUNTIF(S)
verduideliking
Almal ken die gewone COUNTIF funksie – dit laat mens toe om te tel hoeveel selle aan ’n sekere kriteria voldoen. Die COUNTIFS-funksie tel eenvoudig die aantal selle wat aan meer as een kriteria voldoen. 
Kom ons kyk na ’n voorbeeld. Ons wil weet hoeveel mense by die Bloubosrand-tak ouer as 40 jaar is. 
	[image: ]
	Die argumente vir hierdie funksie is pare van CRITERIA RANGES (watter selle om te ondersoek) en CRITERIA (waarvoor om in elke reeks te soek). Jy kan aanhou om reeks/kriteria pare (range/criteria pairs), tot ’n maksimum van 127 by te voeg. Die eerste reeks/kriteria paar is verpligtend, die res is opsioneel. 
[image: ]


Let Wel: Onthou dat elke ekstra reeks dieselfde aantal rye en kolomme as die eerste reeks moet hê (bv, d2:d39 en f2:f39 – kan nie gaan tot f40.) Dit is van toepassing op countifs, sumif(s) & averageif(s).
	’n Ander voorbeeld van ’n COUNTIFS-funksie kan wees om uit te vind hoeveel mense tussen 40 en 50 jaar oud is.
[image: ]
	[image: ]


Let Wel: Dit mag makliker wees om die function builder by 'n COUNTIF te gebruik, omdat aanhalingstekens outomaties bygevoeg word waar dit nodig is. 
AKTIWITEIT 16
1. Voeg funksies in selle B43 tot B46 in om te bepaal hoeveel mans in elke afdeling (department) werk. 
· Maklikste (want jy kan dit na onder kopieer) =COUNTIFS(B$2:B$39,"M",E$2:E$39,A43) of =COUNTIFS($B$2:$B$39,"M",$E$2:$E$39,A43)
· Moeiliker (omdat die argumente vir elke sel aangepas moet word) =COUNTIFS(B2:B39,"M",E2:E39,"Management")
2. Voeg funksies in selle B48 tot B51 in om te bepaal hoeveel vroue in elke afdeling (department) werk .  Soos hierbo.
3. Voeg funksies in selle B54 tot B65 in om te bepaal hoeveel mans by elke tak werk.
· Maklikste (want jy kan dit na onder kopieer) =COUNTIFS(B$2:B$39,"M",F$2:F$39,A54) of =COUNTIFS($B$2:$B$39,"M",$F$2:$F$39,A54)
· Moeiliker (omdat die argumente vir elke sel aangepas moet word) =COUNTIFS(B2:B39,"M",F2:F39,"Blairgowrie")
4. Voeg funksies in selle D54 tot D65 in om te bepaal hoeveel vroue by elke tak werk. Soos hierbo.
Wenk: gebruik altyd absolute verwysings om die reeks van Countif(s), Sumif(s) en Averageif(s)-funksies te 'lock'.
[bookmark: _Toc456196326][bookmark: _Toc458453770]SUMIF(S)
VERDUIDELIKING
Die gewone SUMIF is een van die moeiliker funksies, maar dit help om die function builder te gebruik en mooi op die sintaks te let. SUMIF word gebruik om die waardes in ’n reeks selle (die SUM RANGE) op te tel wat aan ’n enkele reeks of kriteria-paar voldoen. 
	’n Voorbeeld van ’n gewone SUMIF sou wees om die salaris van al die mans op te tel:
	[image: ]


Let Wel: Dit mag help om te onthou dat die eerste twee argumente van die gewone SUMIF dieselfde is as dié van die COUNTIF. Die derde argument Sum_range word dan net aan die einde bygevoeg.
	Die SUMIFS-funksie verskil van die COUNTIFS-funksie, met dié dat dit met die Sum_range (die getalle wat opgetel moet word) begin, en gevolg word deur een of meer reeks/kriteria-pare, tot ’n maksimum van 127. In hierdie voorbeeld het ons die salarisse van al die mans bo 40 jaar opgetel.
	[image: ]


AKTIWITEIT 17
1. Voeg funksies in selle C43 tot C46 in om die totale salarisse van mans in elke aparte department te bereken.  
· Maklikste (want jy kan dit na onder kopieer) =SUMIFS(G$2:G$39,B$2:B$39,"M",E$2:E$39,A43) of =SUMIFS($G$2:$G$39,$B$2:$B$39,"M",$E$2:$E$39,A43)
· Moeiliker (omdat die argumente vir elke sel aangepas moet word) 
=SUMIFS(G2:G39,B2:B39,"M",E2:E39,A43)
2. Voeg funksies in selle C48 tot C51 in om die totale salarisse van vrouens in elke aparte department te bereken. Soos hierbo.
3. Voeg funksies in selle C54 tot C65 in om die totale salarisse van mans by elke afsonderlike tak te bereken. 
· Maklikste (want jy kan dit na onder kopieer) =SUMIFS(G$2:G$39,B$2:B$39,"M",F$2:F$39,A54) or =SUMIFS($G$2:$G$39,$B$2:$B$39,"M",$F$2:$F$39,A54)
· Moeiliker (omdat die argumente vir elke sel aangepas moet word) 
=SUMIFS(G2:G39,B2:B39,"M",F2:F39,A54)
[bookmark: _Toc457647147][bookmark: _Toc458269504]

[bookmark: _Toc458453771]AVERAGEIF(S) [nie in die eksamenriglyne gelys nie, maar ‘n nuttige funksie]
VERDUIDELIKING
Die AVERAGEIF-funksie is baie nuttig wanneer jy die gemiddelde waarde wil bereken van selle wat aan ’n enkele kriterium (voorwaarde) voldoen. ’n Onderwyser kan byvoorbeeld hierdie funksie gebruik om die gemiddelde punt van leerders vir ’n toets te bepaal, uitsluitend diegene wat afwesig was. 
	[image: ]
	In hierdie voorbeeld is die ‘range’ en die ‘Average_range’ dieselfde.



	Kom ons kyk na ’n ander voorbeeld. Gestel ons wil die gemiddelde ouderdom van die mense in die bestuursafdeling (Management department) bepaal. 
In hierdie geval kom die ‘range’ en ‘criteria’ eerste, soos by ’n COUNTIF. Dan word die reeks bygevoeg waarvan jy die gemiddeld wil hê (die ‘Average_range’) as die derde argument.
	[image: ]



	Die sintaksis van die AVERAGEIFS-funksie is soortgelyk aan dié van ’n SUMIFS, beginnende met die ‘Average_range’; daarna word jy toegelaat om verdere reeks/kriteria-pare by te voeg, tot ’n maksimum van 127. In hierdie voorbeeld bereken ons die gemiddelde ouderdom van mans met ’n salaris van meer as R20 000.
	[image: ]


AKTIWITEIT 18
1. Voeg ’n funksie in sel G41 in om die gemiddelde salaris van al die manlike werknemers te bepaal. =AVERAGEIF(B2:B39,"M",G2:G39)
2. Voeg ’n funksie in sel G42 in om die gemiddelde ouderdom van alle werknemers wie se name met ’n N begin, te bereken. =AVERAGEIF(A2:A39,"N*",D2:D39)
3. Voeg ’n funksie in sel I41 in om die gemiddelde salaris van mans in die Verkope (Sales)-afdeling, te bereken. =AVERAGEIFS(G2:G39,B2:B39,"M",E2:E39,"Sales")
4. Voeg ’n funksie in sel I42 in om die gemiddelde ouderdom van vroue wat by die Lanseria-tak werk, te bereken.  =AVERAGEIFS(D2:D39,B2:B39,"F",F2:F39,"Lanseria")
Let op: Moet nooit kriteria spesifiseer deur na selle in die 'rou' data te verwys nie. In vraag 1 hierbo, word bv met die datareeks A2:H39 gewerk. Die funksie =AVERAGEIF(B2:B39,B38,G2:G39) is dus onaanvaarbaar. Dit is omdat die kriterium wat in die tweede argument van die funksie genoem word  (B38) – alhoewel dit tans “M” vir manlik is –  nie vas is nie, maar dit kan trouens verander (dit mag selfs verkeerd ingetik wees). In hierdie geval, as daar nie 'n "M" op sy eie in 'n sel buite die datareeks, bv as 'n opskrif of etiket is nie, gebruik die waarde in aanhalingstekens – "M".


[bookmark: _Toc458453772]LOGIESE FUNKSIES
Alhoewel hierdie funksies nie spesifiek in die eksamenriglyne genoem word nie, kan hulle baie nuttig wees wanneer dit met die IF-funksie gekombineer word. Sodoende word die nodigheid om ekstra kolomme vir tydelike bewerkings te skep (of ander omslagtige metodes) uitgeskakel.
Maak die sigblad ExcelExercises oop en werk in die Logical Functions-werkblad.
[bookmark: _Toc458453773]EENVOUDIGE LOGIESE TOETS
VERDUIDELIKING
	[image: ]
	’n IF-funksie voer eers ’n logiese toets uit, en in plaas daarvan om net eenvoudig TRUE of FALSE as antwoord te gee, stel dit jou in staat om ’n waarde in die plek van die TRUE of FALSE te gee. 
 [image: ]


’n Eenvoudige logiese toets kan egter ook op sy eie (buite ’n IF-funksie) uitgevoer word.  
In ’n logiese toets, toets mens ’n enkele sel met ’n voorwaarde deur gebruik te maak van die operators = , >, <, of ’n kombinasie van hulle. Kom ons kyk na ’n paar voorbeelde.  
	[image: ]
	Die eerste  = is maar net daar om ’n formule (of ’n funksie) mee te begin en is nie deel van die logiese toets self nie. In hierde voorbeeld toets ons of B2 gelyk is aan F. Die antwoord sal  bloot TRUE (waar) of FALSE (onwaar) wees.



	[image: ]
	Die formule in hierdie voorbeeld toets of die waarde in C2 (die persoon se ouderdom) bo 60 is en dit gee weereens net ’n eenvoudige TRUE of FALSE.

1. Toets in sel H2 of die werknemer manlik is. Die toets moet TRUE vertoon indien die persoon manlik is en FALSE indien die persoon vroulik is. Kopieer die formule na die ander werknemers. =B2="M"
2. Toets in sel I2 of die werknemer 25 jaar of jonger is. TRUE moet in die sel verskyn as sy ouderdom 25 of minder is en FALSE as die getal groter as dit is. Kopieer die formule na die ander werknemers. =C2<=25
[bookmark: _Toc456182945][bookmark: _Toc456196320][bookmark: _Toc458453774]OR-FUNKSIE
VERDUIDELIKING
Hierdie funksie toets twee voorwaardes – indien enige een of altwee  van die voorwaardes WAAR is, sal daar TRUE verskyn. Indien altwee voorwaardes FALSE is, sal daar FALSE verskyn. Kom ons kyk na ’n voorbeeld. 
Veronderstel die maatskappy wil al die werknemers in die Verkope-afdeling (Sales department), sowel as al die werknemers van die Fairland-tak op ’n kursus stuur. Vir hierdie scenario sal mens die OR-funksie gebruik.
	[image: ]
	In hierdie voorbeeld is die eerste logiese toets of die department ‘Sales’ is, en die tweede of hulle by die Fairland-tak werk. As enige een of altwee van hierdie voorwaardes waar is, sal die antwoord TRUE wees. Let op die sintaksis: die logiese toetse word deur ’n komma geskei. Slegs die eerste logiese toets is verpligtend, daarna kan jy soveel as 255 toetse byvoeg!


Let Wel: ’n opsionele deel van ’n funksie word deur blokhakies [] in die tooltip aangedui.

1. Toets in sel J2 of die werknemer vroulik, of ouer as 40 is. Kopieer hierdie funksie na die ander werknemers. =OR(B2="F",C2>40) Let daarna op dat die funksies in selle  J4 (vroulik, jonger as 40) en J10 (manlik, ouer as 40) altwee TRUE vertoon.
2. Toets in sel K2 of die werknemer se salaris minder as R10 000 is, of dat sy/haar department die werkswinkel (Workshop) is. Kopieer hierdie funksie na die ander deelnemers toe. =OR(F2<10000,D2="Workshop") Let op dat die funksies in selle K2 (verdien minder as R10 000 en werk in die Workshop), K14 (verdien minder as R10 000 maar werk by Spare Parts) en K15 (werk in die Workshop maar verdien meer as R10 000) amal TRUE vertoon.
[bookmark: _Toc456182946][bookmark: _Toc456196321][bookmark: _Toc458453775]AND-FUNKSIE
VERDUIDELIKING
Hierdie funksie toets ook twee voorwaardes – as altwee voorwaardes WAAR is, sal die funksie TRUE vertoon. As enige een of altwee van die voorwaardes ONWAAR is, sal die funksie FALSE vertoon.  Kom ons kyk na ’n voorbeeld. 
Veronderstel die werknemers wat in die werkswinkel werk en wat R10 000 of minder verdien, het ’n protesoptog geloods en moet nou ’n salarisverhoging ontvang.  
	[image: ]
	In hierdie voorbeeld toets ons of die werknemer in die Workshop werk EN (AND) R10 000 of minder verdien. Slegs as ALTWEE hierdie voorwaardes waar is, sal die antwoord TRUE wees. Die sintaks is dieselfde as vir die OR-funksie.

1. Toets in sel L2 of die werknemer 35 jaar of jonger is EN by die Randpark Ridge-tak werk. Kopieer hierdie funksie na die ander werknemers toe.=AND(C2<=35,E2="Randpark Ridge") Let op dat die funksie in sel L2 (werk by Randpark Ridge, maar ouer as 35) FALSE is. Slegs die funksies in selle L5 en L8 voldoen aan altwee voorwaardes en gee TRUE as antwoord.
2. Toets in sel M2 of die werknemer se salaris meer as R15 000 is en sy/haar department Management is. Kopieer hierdie funksie na die ander werknemers toe. =AND(D2="Management",F2>15000) Let op dat die funksie in sell M7 FALSE is, selfs al verdien hy meer as R15 000 (want hy werk in ‘Sales’, nie in ‘Management’ nie). Die antwoorde in kolom M is slegs TRUE as altwee die voorwaardes waar is.
[bookmark: _Toc456182947][bookmark: _Toc456196322][bookmark: _Toc458453776]KOMBINEER OR-FUNKSIE MET IF
VERDUIDELIKING
’n OR-funksie kan baie nuttig wees wanneer dit met ’n IF-funksie gekombineer word en kan in plaas van ’n geneste IF (wat ’n tweede IF in die Value_If_false-boksie gebruik), gebruik word. Kom ons gebruik een van die vorige voorbeelde om dit te illustreer.
Veronderstel die maatskappy wil al die werknemers in die Verkope-afdeling (Sales department), sowel as al die werknemers van die Fairland-tak, op ’n kursus stuur. Die woord ‘Course’ moet in die toepaslike sel verskyn indien een of altwee hierdie voorwaardes waar is, anders moet die sel leeg gelaat word:
	[image: ]
	Soos jy kan sien, is daar in plaas van ’n enkele logiese toets in die IF-funksie, van ’n OR- funksie gebruik gemaak. Dieselfde antwoord sou verkry word deur ’n geneste IF te gebruik.

	[image: ]

1. Die maatskappy het agtergekom dat daar baie klagtes oor die personeel van die Ferndale-tak is. Hulle het besluit om ’n opknapppingskursus in kliëntediens vir al die werknemers van die Ferndale-tak, sowel as al die werknemers wat na 2013 begin werk het, aan te bied.
Voeg ’n funksie in sel N2 in wat die teks ‘Customer Service Training’ sal vertoon indien Andrew by die Ferndale-tak werk of na 2012 by die maatskappy begin werk het. As nie een van hierdie twee voorwaardes geld nie, moet die sel leeg vertoon. Kopieer die funksie na die ander werknemers toe. 
=IF(OR(E2="Ferndale",G2>"2012-12-31"),"Customer Service Training","")
Let op: ’n datum kan ook as ’n kriterium gebruik word – formateer dit eenvoudig dieselfde as die rekenaar se Short Date, bv. >"1990/01/01". Dit is noodsaaklik om die aanhalingstekens om die datum te voeg.
[bookmark: _Toc456182948][bookmark: _Toc456196323][bookmark: _Toc458453777]KOMBINEER AND-FUNKSIE MET WITH IF
VERDUIDELIKING
Die AND-funksie kan net so nuttig wees wanneer dit met ’n IF-funksie gekombineer word en kan gebruik word om ’n geneste IF te vervang. Kom ons gebruik ’n vorige voorbeeld (bespreek onder die AND-funksie) en kombineer dit met ’n IF-funksie. 
	[image: ]
	Veronderstel werknemers in die Workshop wat R10 000 of minder verdien, het ’n protesoptog geloods en moet ’n salarisverhoging van 10% ontvang. Bereken en vertoon hulle nuwe salaris indien aan albei voorwaardes voldoen word, of wys hulle huidige salaris indien daar nie aan een of albei voorwaardes voldoen word nie. Jou funksie sal soos volg lyk (let daarop dat die tweede logiese toets bietjie afgekap is – dit moet wees F2<=10000). Slegs indien altwee hierdie voorwaardes waar is, word die 10% verhoging toegepas.


Let Wel: Jy kan ook tik F2*110% of F2+F2*0.1 in die Value_if_true-boksie.
	Die geneste alternatief, sal so lyk:
	[image: ]

1. Werknemers in die ‘Management department’ wie se salaris tussen R15 000 en R20 000 is, sal ’n prestasiebonus van R2 000 ontvang wat by hulle salaris gevoeg sal word. Indien werknemers aan nie een van hierdie voorwaardes voldoen nie, sal hulle salaris onveranderd bly.
Voeg ’n funksie in sel O2 in om die bedrag wat Andrew sal ontvang, te bereken. Kopieer hierdie funksie na die ander werknemers. 
=IF(AND(D2="Management",F2>=15000,F2<=20000),F2+2000,F2) Let op dat net sel O2 die bygevoegde bonus wys.
2. Gebruik voorwaardelike formatering om aan te dui watter waardes in kolom N groter is as die waardes in kolom F. 
Oplossing: pas voorwaardelike formatering toe op selle O2:O39 – kies Highlight Cell Rules  Greater Than... en klik op sel F2. Verwyder die absolute selverwysing van die formule in die boks. (Dit mag dalk makliker wees om die inhou van die boks te verwyder en =F2 in te tik). Dit laat al die verhelderde selle in kolom O toe om met hul ooreenstemmende selle in kolom F te vergelyk

